» Data Source: ns1_zone

Provides details about a NS1 Zone. Use this if you would simply like to read information from NS1 into your configurations. For read/write operations, you should use a resource.

» Example Usage

```
# Get details about a NS1 Zone.
data "ns1_zone" "example" {
  zone = "terraform.example.io"
}
```

» Argument Reference

- `zone` - (Required) The domain name of the zone.

» Attributes Reference

In addition to the argument above, the following are exported:

- `link` - The linked target zone.
- `primary` - The primary zones' IPv4 address.
- `additional_primaries` - List of additional IPv4 addresses for the primary zone.
- `ttl` - The SOA TTL.
- `refresh` - The SOA Refresh.
- `retry` - The SOA Retry.
- `expiry` - The SOA Expiry.
- `nx_ttl` - The SOA NX TTL.
- `dnssec` - Whether or not DNSSEC is enabled for the zone.
- `networks` - List of network IDs for which the zone is available.
- `dns_servers` - Authoritative Name Servers.
- `hostmaster` - The SOA Hostmaster.
- `secondaries` - List of secondary servers. Secondaries is documented below.

» Secondaries

A secondary has the following fields:

- `ip` - IPv4 address of the secondary server.
- `port` - Port of the the secondary server. Default `53`.

- `notify` - Whether we send `NOTIFY` messages to the secondary host when the zone changes. Default `false`.
- `networks` - List of network IDs (`int`) for which the zone should be made available. Default is network 0, the primary NSONE Global Network.

» Data Source: ns1_dnssec

Provides DNSSEC details about a NS1 Zone.

» Example Usage

```
# Get DNSSEC details about a NS1 Zone.
resource "ns1_zone" "example" {
  zone   = "terraform.example.io"
  dnssec = true
}

data "ns1_dnssec" "example" {
  zone = "${ns1_zone.example.zone}"
}
```

» Argument Reference

- `zone` - (Required) The name of the zone to get DNSSEC details for.

» Attributes Reference

In addition to the argument above, the following are exported:

- `keys` - (Computed) - Keys field is documented below.
- `delegation` - (Computed) - Delegation field is documented below.

» Keys

`keys` has the following fields:

- `dnskey` - (Computed) List of Keys. Key is documented below.
- `ttl` - (Computed) TTL for the Keys (int).

» Delegation

`delegation` has the following fields:

- `dnskey` - (Computed) List of Keys. Key is documented below.
- `ds` - (Computed) List of Keys. Key is documented below.
- `ttl` - (Computed) TTL for the Keys (int).

» Key

A `key` has the following (string) fields:

- `flags` - (Computed) Flags for the key.
- `protocol` - (Computed) Protocol of the key.
- `algorithm` - (Computed) Algorithm of the key.
- `public_key` - (Computed) Public key for the key.

» ns1_zone

Provides a NS1 DNS Zone resource. This can be used to create, modify, and delete zones.

» Example Usage

```
# Create a new DNS zone
resource "ns1_zone" "example" {
  zone = "terraform.example.io"
  ttl  = 600
}

# Create a new primary zone
resource "ns1_zone" "example_primary" {
  zone     = "terraform-primary.example.io"
  secondaries {
    ip     = "2.2.2.2"
  }
  secondaries {
    ip     = "3.3.3.3"
    port   = 5353
    notify = true
  }
}

# Create a new secondary zone
```

```
resource "ns1_zone" "example_primary" {
  zone     = "terraform-primary.example.io"
  primary  = "2.2.2.2"
  additional_primaries = ["3.3.3.3", "4.4.4.4"]
}

# Create a zone with NS record managed by terraform
resource "ns1_zone" "example_ns_management" {
  zone                   = "terraform-ns.example.io"
  autogenerate_ns_record = false
}
# Use the dns_servers attribute to set answers for our NS1-provided nameservers
resource "ns1_record" "example_ns_management_ns" {
  zone   = ns1_zone.example_ns_management.zone
  domain = ns1_zone.example_ns_management.zone
  type   = "NS"

  dynamic "answers" {
    for_each = split(",", ns1_zone.example_ns_management.dns_servers)
    content {
      answer = answers.value
    }
  }

  answers {
    answer = "some_other_nameserver.example.com"
  }
}
```

» Argument Reference

The following arguments are supported:

- `zone` - (Required) The domain name of the zone.
- `link` - (Optional) The target zone(domain name) to link to.
- `primary` - (Optional) The primary zones' IPv4 address. This makes the zone a secondary. Conflicts with `secondaries`.
- `additional_primaries` - (Optional) List of additional IPv4 addresses for the primary zone. Conflicts with `secondaries`.
- `ttl` - (Optional/Computed) The SOA TTL.
- `refresh` - (Optional/Computed) The SOA Refresh. Conflicts with `primary` and `additional_primaries` (default must be accepted).
- `retry` - (Optional/Computed) The SOA Retry. Conflicts with `primary` and `additional_primaries` (default must be accepted).
- `expiry` - (Optional/Computed) The SOA Expiry. Conflicts with `primary`

and `additional_primaries` (default must be accepted).
- `nx_ttl` - (Optional/Computed) The SOA NX TTL. Conflicts with `primary` and `additional_primaries` (default must be accepted).
- `dnssec` - (Optional/Computed) Whether or not DNSSEC is enabled for the zone. Note that DNSSEC must be enabled on the account by support for this to be set to `true`.
- `networks` - (Optional/Computed) List of network IDs for which the zone is available. If no network is provided, the zone will be created in network 0, the primary NS1 Global Network.
- `secondaries` - (Optional) List of secondary servers. This makes the zone a primary. Conflicts with `primary` and `additional_primaries`. Secondaries is documented below.
- `autogenerate_ns_record` - (Optional, default true). If set to false, clears the autogenerated NS record on zone creation. This allows an automated workflow for creating zones with the NS record in terraform state. See above for an example. Note that this option only has an effect when a zone is being created.

» Secondaries

A zone can have zero or more `secondaries`. Note how this is implemented in the example above. A secondary has the following fields:

- `ip` - (Required) IPv4 address of the secondary server.
- `port` - (Optional) Port of the the secondary server. Default `53`.
- `notify` - (Optional) Whether we send `NOTIFY` messages to the secondary host when the zone changes. Default `false`.
- `networks` - (Computed) - List of network IDs (`int`) for which the zone should be made available. Default is network 0, the primary NSONE Global Network. Normally, you should not have to worry about this.

» Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `dns_servers` - (Computed) Authoritative Name Servers.
- `hostmaster` - (Computed) The SOA Hostmaster.

» A note on making Primary or Secondary changes to zones

Switching a zone to being a secondary forces a new resource. In other words, the zone will first be destroyed, then recreated as a secondary. Editing or removing the `primary` key, or directly changing a secondary zone to a primary (by removing the `primary` and `additional_primaries` keys, and setting `secondaries`) is supported ”in place”. However, in these situations we do not alter records on

the zone. You may need to amend records, or finagle them into Terraform state. As a particular example, if you change a secondary zone to be primary (or just not-a-secondary) before a zone transfer has occurred, you can end up with no records on the zone.

Currently, this provider does not support zones being both Primary and Secondary. If that functionality is important for your workflow, please open an issue or contact support, so we can prioritize the work accordingly.

» Import

```
terraform import ns1_zone.<name> <zone>
```

So for the example above:

```
terraform import ns1_zone.example terraform.example.io
```

» ns1_record

Provides a NS1 Record resource. This can be used to create, modify, and delete records.

» Example Usage

```
resource "ns1_zone" "example" {
  zone = "terraform.example.io"
}

resource "ns1_datasource" "ns1" {
  name       = "ns1_source"
  sourcetype = "nsone_v1"
}

resource "ns1_datafeed" "foo" {
  name      = "foo_feed"
  source_id = "${ns1_datasource.ns1.id}"
  config = {
    label = "foo"
  }
}

resource "ns1_datafeed" "bar" {
  name      = "bar_feed"
  source_id = "${ns1_datasource.ns1.id}"
```

```
  config = {
    label = "bar"
  }
}

resource "ns1_record" "www" {
  zone   = "${ns1_zone.tld.zone}"
  domain = "www.${ns1_zone.tld.zone}"
  type   = "CNAME"
  ttl    = 60
  meta   = {
    up   = true
  }

  regions {
    name = "east"
    meta = {
      georegion = "US-EAST"
    }
  }

  regions {
    name = "usa"
    meta = {
      country = "US"
    }
  }

  answers {
    answer  = "sub1.${ns1_zone.tld.zone}"
    region  = "east"
    meta    = {
      up    = "{\"feed\":\"${ns1_datafeed.foo.id}\"}"
    }
  }

  answers {
    answer = "sub2.${ns1_zone.tld.zone}"
    meta   = {
      up   = "{\"feed\":\"${ns1_datafeed.bar.id}\"}"
      connections = 3
    }
  }

  # Example of setting pulsar metadata on an answer. Note the use of
  # jsonencode (available in terraform 0.12+). This is preferable to the
```

```
  # "quoted JSON" style used for feeds above, both for readability, and
  # because it handles ordering issues as well.
  answers {
    answer = "sub3.${ns1_zone.tld.zone}"
    meta   = {
      pulsar = jsonencode([{
        "job_id"     = "abcdef",
        "bias"       = "*0.55",
        "a5m_cutoff" = 0.9
      }])
    }
  }

  filters {
    filter = "select_first_n"

    config = {
      N = 1
    }
  }
}

# Some other non-NS1 provider that returns a zone with a trailing dot and a domain with a le
resource "external_source" "baz" {
  zone      = "terraform.example.io."
  domain    = ".www.terraform.example.io"
}

# Basic record showing how to clean a zone or domain field that comes from
# another non-NS1 resource. DNS names often end in '.' characters to signify
# the root of the DNS tree, but the NS1 provider does not support this.
#
# In other cases, a domain or zone may be passed in with a preceding dot ('.')
# character which would likewise lead the system to fail.
resource "ns1_record" "external" {
  zone   = replace("${external_source.zone}", "/(^\\.)|(\\.$)/", "")
  domain = replace("${external_source.domain}", "/(^\\.)|(\\.$)/", "")
  type   = "CNAME"
}
```

» Argument Reference

The following arguments are supported:

- `zone` - (Required) The zone the record belongs to. Cannot have leading

or trailing dots (".") - see the example above.

- `domain` - (Required) The records' domain. Cannot have leading or trailing dots - see the example above.
- `type` - (Required) The records' RR type.
- `ttl` - (Optional) The records' time to live.
- `link` - (Optional) The target record to link to. This means this record is a 'linked' record, and it inherits all properties from its target.
- `use_client_subnet` - (Optional) Whether to use EDNS client subnet data when available(in filter chain).
- `meta` - (Optional) meta is supported at the `record` level. Meta is documented below.
- `regions` - (Optional) One or more "regions" for the record. These are really just groupings based on metadata, and are called "Answer Groups" in the NS1 UI, but remain `regions` here for legacy reasons. Regions are documented below. Please note the ordering requirement!
- `answers` - (Optional) One or more NS1 answers for the records' specified type. Answers are documented below.
- `filters` - (Optional) One or more NS1 filters for the record(order matters). Filters are documented below.

» Answers

`answers` support the following:

- `answer` - (Required) Space delimited string of RDATA fields dependent on the record type.

  A:

  ```
  answer = "1.2.3.4"
  ```

  CNAME:

  ```
  answer = "www.example.com"
  ```

  MX:

  ```
  answer = "5 mail.example.com"
  ```

  SRV:

  ```
  answer = "10 0 2380 node-1.example.com"
  ```

  SPF:

  ```
  answer = "v=DKIM1; k=rsa; p=XXXXXXXX"
  ```

- `region` - (Optional) The region (Answer Group really) that this answer belongs to. This should be one of the names specified in `regions`. Only a single `region` per answer is currently supported. If you want an answer

in multiple regions, duplicating the answer (including metadata) is the correct approach.

- `meta` - (Optional) meta is supported at the `answer` level. Meta is documented below.

» Filters

`filters` support the following:

- `filter` - (Required) The type of filter.
- `disabled` - (Optional) Determines whether the filter is applied in the filter chain.
- `config` - (Optional) The filters' configuration. Simple key/value pairs determined by the filter type.

» Regions

`regions` support the following:

- `name` - (Required) Name of the region (or Answer Group).
- `meta` - (Optional) meta is supported at the `regions` level. Meta is documented below. Note that `Meta` values for `country`, `ca_province`, `georegion`, and `us_state` should be comma separated strings, and changes in ordering will not lead to terraform detecting a change.

Note: regions **must** be sorted lexically by their "name" argument in the Terraform configuration file, otherwise Terraform will detect changes to the record when none actually exist.

» Meta

Metadata (`meta`) is a bit tricky at the moment. For "static" values it works as you would expect, but when a value is a `datafeed`, or a JSON object, it needs some tweaks to work correctly.

If using terraform 0.12+, we can use the `jsonencode` function (see the Example Usage above). This handles translating the field to and from strings as needed, and also handles ordering issues that otherwise may need to be handled manually, in config or by the provider.

If you are NOT using terraform 0.12, these values should be represented as "escaped" JSON. See the Example Usage above for illustration of this. Note that variables are still supported in the escaped JSON format.

Since this resource supports import, you may find it helpful to set up some `meta` fields via the web portal or API, and use the results from import to check your syntax and ensure that everything is properly escaped and evaluated.

See NS1 API for the most up-to-date list of available `meta` fields.

» FQDN Formatting

Different providers may have different requirements for FQDN formatting. A common thing is to return or require a trailing dot, e.g. foo.com. The NS1 provider does not require or support trailing or leading dots, so depending on what resources you are connecting, a little bit of replacement might be needed. See the example above.

» Attributes Reference

All of the arguments listed above are exported as attributes, with no additions.

» Import

```
terraform import ns1_record.<name> <zone>/<domain>/<type>
```

So for the example above:

```
terraform import ns1_record.www terraform.example.io/www.terraform.example.io/CNAME
```

» ns1_monitoringjob

Provides a NS1 Monitoring Job resource. This can be used to create, modify, and delete monitoring jobs.

» Example Usage

```
resource "ns1_monitoringjob" "uswest_monitor" {
  name          = "uswest"
  active        = true
  regions       = ["sjc", "sin", "lga"]
  job_type      = "tcp"
  frequency     = 60
  rapid_recheck = true
  policy        = "quorum"

  config = {
    send = "HEAD / HTTP/1.0\r\n\r\n"
    port = 80
    host = "example-elb-uswest.aws.amazon.com"
```

```
  }

  rules {
    value      = "200 OK"
    comparison = "contains"
    key        = "output"
  }
}
```

» Argument Reference

The following arguments are supported:

- `name` - (Required) The free-form display name for the monitoring job.
- `job_type` - (Required) The type of monitoring job to be run. See NS1 API docs for supported values.
- `active` - (Required) Indicates if the job is active or temporarily disabled.
- `regions` - (Required) The list of region codes in which to run the monitoring job. See NS1 API docs for supported values.
- `frequency` - (Required) The frequency, in seconds, at which to run the monitoring job in each region.
- `rapid_recheck` - (Required) If true, on any apparent state change, the job is quickly re-run after one second to confirm the state change before notification.
- `policy` - (Required) The policy for determining the monitor's global status based on the status of the job in all regions. See NS1 API docs for supported values.
- `config` - (Required) A configuration dictionary with keys and values depending on the jobs' type.
- `notify_delay` - (Optional) The time in seconds after a failure to wait before sending a notification.
- `notify_repeat` - (Optional) The time in seconds between repeat notifications of a failed job.
- `notify_failback` - (Optional) If true, a notification is sent when a job returns to an "up" state.
- `notify_regional` - (Optional) If true, notifications are sent for any regional failure (and failback if desired), in addition to global state notifications.
- `notify_list` - (Optional) The id of the notification list to send notifications to.
- `notes` - (Optional) Freeform notes to be included in any notifications about this job.
- `rules` - (Optional) A list of rules for determining failure conditions. Job Rules are documented below.

Monitoring Job Rules (`rules`) support the following:

- `key` - (Required) The output key.
- `comparison` - (Required) The comparison to perform on the the output.
- `value` - (Required) The value to compare to.

» Attributes Reference

All of the arguments listed above are exported as attributes, with no additions.

» ns1_notifylist

Provides a NS1 Notify List resource. This can be used to create, modify, and delete notify lists.

» Example Usage

```
resource "ns1_notifylist" "nl" {
  name = "my notify list"
  notifications {
    type = "webhook"
    config = {
      url = "http://www.mywebhook.com"
    }
  }

  notifications {
    type = "email"
    config = {
      email = "test@test.com"
    }
  }
}
```

» Argument Reference

The following arguments are supported:

- `name` - (Required) The free-form display name for the notify list.
- `notifications` - (Optional) A list of notifiers. All notifiers in a notification list will receive notifications whenever an event is send to the list (e.g., when a monitoring job fails). Notifiers are documented below.

Notify List Notifiers (`notifications`) support the following:

- `type` - (Required) The type of notifier. Available notifiers are indicated in /notifytypes endpoint.
- `config` - (Required) Configuration details for the given notifier type.

» Attributes Reference

All of the arguments listed above are exported as attributes, with no additions.

» ns1_datasource

Provides a NS1 Data Source resource. This can be used to create, modify, and delete data sources.

» Example Usage

```
resource "ns1_datasource" "example" {
  name       = "example"
  sourcetype = "nsone_v1"
}
```

» Argument Reference

The following arguments are supported:

- `name` - (Required) The free form name of the data source.
- `sourcetype` - (Required) The data sources type, listed in API endpoint https://api.nsone.net/v1/data/sourcetypes.
- `config` - (Optional) The data source configuration, determined by its type, matching the specification in `config` from /data/sourcetypes.

» Attributes Reference

All of the arguments listed above are exported as attributes, with no additions.

» ns1_datafeed

Provides a NS1 Data Feed resource. This can be used to create, modify, and delete data feeds.

» Example Usage

```
resource "ns1_datasource" "example" {
  name       = "example"
  sourcetype = "nsone_v1"
}

resource "ns1_datafeed" "uswest_feed" {
  name      = "uswest_feed"
  source_id = "${ns1_datasource.example.id}"

  config = {
    label = "uswest"
  }
}

resource "ns1_datafeed" "useast_feed" {
  name      = "useast_feed"
  source_id = "${ns1_datasource.example.id}"

  config = {
    label = "useast"
  }
}
```

» Argument Reference

The following arguments are supported:

- `source_id` - (Required) The data source id that this feed is connected to.
- `name` - (Required) The free form name of the data feed.
- `config` - (Optional) The feeds configuration matching the specification in `feed_config` from /data/sourcetypes.

» Attributes Reference

All of the arguments listed above are exported as attributes, with no additions.

» ns1_apikey

Provides a NS1 Api Key resource. This can be used to create, modify, and delete api keys.

» Example Usage

```
resource "ns1_team" "example" {
  name = "Example team"
}

resource "ns1_apikey" "example" {
  name  = "Example key"
  teams = ["${ns1_team.example.id}"]

  # Optional IP whitelist
  ip_whitelist = ["1.1.1.1","2.2.2.2"]

  # Configure permissions
  dns_view_zones       = false
  account_manage_users = false
}
```

» Permissions

An API key will inherit permissions from the teams it is assigned to. If a key is assigned to a team and also has individual permissions set on the key, the individual permissions will be overridden by the inherited team permissions. In a future release, setting permissions on a key that is part of a team will be explicitly disabled.

When a key is removed from all teams completely, it will inherit whatever permissions it had previously. If a key is removed from all it's teams, it will probably be necessary to run `terraform apply` a second time to update the keys permissions from it's old team permissions to new key-specific permissions.

See the NS1 API docs for an overview of permission semantics.

» Argument Reference

The following arguments are supported:

- `name` - (Required) The free form name of the apikey.
- `key` - (Required) The apikeys authentication token.
- `teams` - (Optional) The teams that the apikey belongs to.
- `ip_whitelist` - (Optional) The IP addresses to whitelist for this key.
- `ip_whitelist_strict` - (Optional) Sets exclusivity on this IP whitelist.
- `dns_view_zones` - (Optional) Whether the apikey can view the accounts zones.

- `dns_manage_zones` - (Optional) Whether the apikey can modify the accounts zones.
- `dns_zones_allow_by_default` - (Optional) If true, enable the `dns_zones_allow` list, otherwise enable the `dns_zones_deny` list.
- `dns_zones_allow` - (Optional) List of zones that the apikey may access.
- `dns_zones_deny` - (Optional) List of zones that the apikey may not access.
- `data_push_to_datafeeds` - (Optional) Whether the apikey can publish to data feeds.
- `data_manage_datasources` - (Optional) Whether the apikey can modify data sources.
- `data_manage_datafeeds` - (Optional) Whether the apikey can modify data feeds.
- `account_manage_users` - (Optional) Whether the apikey can modify account users.
- `account_manage_payment_methods` - (Optional) Whether the apikey can modify account payment methods.
- `account_manage_plan` - (Optional) Whether the apikey can modify the account plan.
- `account_manage_teams` - (Optional) Whether the apikey can modify other teams in the account.
- `account_manage_apikeys` - (Optional) Whether the apikey can modify account apikeys.
- `account_manage_account_settings` - (Optional) Whether the apikey can modify account settings.
- `account_view_activity_log` - (Optional) Whether the apikey can view activity logs.
- `account_view_invoices` - (Optional) Whether the apikey can view invoices.
- `monitoring_manage_lists` - (Optional) Whether the apikey can modify notification lists.
- `monitoring_manage_jobs` - (Optional) Whether the apikey can modify monitoring jobs.
- `monitoring_view_jobs` - (Optional) Whether the apikey can view monitoring jobs.
- `security_manage_global_2fa` - (Optional) Whether the apikey can manage global two factor authentication.
- `security_manage_active_directory` - (Optional) Whether the apikey can manage global active directory. Only relevant for the DDI product.
- `dhcp_manage_dhcp` - (Optional) Whether the apikey can manage DHCP. Only relevant for the DDI product.
- `dhcp_view_dhcp` - (Optional) Whether the apikey can view DHCP. Only relevant for the DDI product.
- `ipam_manage_ipam` - (Optional) Whether the apikey can manage IPAM. Only relevant for the DDI product.
- `ipam_view_ipam` - (Optional) Whether the apikey can view IPAM. Only relevant for the DDI product.

» Attributes Reference

All of the arguments listed above are exported as attributes, with no additions.

» ns1_team

Provides a NS1 Team resource. This can be used to create, modify, and delete teams. The credentials used must have the `manage_teams` permission set.

» Example Usage

```
# Create a new NS1 Team
resource "ns1_team" "example" {
  name = "Example team"


  # Optional IP whitelists
  ip_whitelist {
    name = "whitelist-1"
    values = ["1.1.1.1", "2.2.2.2"]
  }
  ip_whitelist {
    name = "whitelist-2"
    values = ["3.3.3.3", "4.4.4.4"]
  }

  # Configure permissions
  dns_view_zones       = false
  account_manage_users = false
}
```

» Argument Reference

The following arguments are supported:

- `name` - (Required) The free form name of the team.
- `ip_whitelist` - (Optional) The IP addresses to whitelist for this key.
- `dns_view_zones` - (Optional) Whether the team can view the accounts zones.
- `dns_manage_zones` - (Optional) Whether the team can modify the accounts zones.
- `dns_zones_allow_by_default` - (Optional) If true, enable the `dns_zones_allow` list, otherwise enable the `dns_zones_deny` list.

- `dns_zones_allow` - (Optional) List of zones that the team may access.
- `dns_zones_deny` - (Optional) List of zones that the team may not access.
- `data_push_to_datafeeds` - (Optional) Whether the team can publish to data feeds.
- `data_manage_datasources` - (Optional) Whether the team can modify data sources.
- `data_manage_datafeeds` - (Optional) Whether the team can modify data feeds.
- `account_manage_users` - (Optional) Whether the team can modify account users.
- `account_manage_payment_methods` - (Optional) Whether the team can modify account payment methods.
- `account_manage_plan` - (Optional) Whether the team can modify the account plan.
- `account_manage_teams` - (Optional) Whether the team can modify other teams in the account.
- `account_manage_apikeys` - (Optional) Whether the team can modify account apikeys.
- `account_manage_account_settings` - (Optional) Whether the team can modify account settings.
- `account_view_activity_log` - (Optional) Whether the team can view activity logs.
- `account_view_invoices` - (Optional) Whether the team can view invoices.
- `monitoring_manage_lists` - (Optional) Whether the team can modify notification lists.
- `monitoring_manage_jobs` - (Optional) Whether the team can modify monitoring jobs.
- `monitoring_view_jobs` - (Optional) Whether the team can view monitoring jobs.
- `security_manage_global_2fa` - (Optional) Whether the team can manage global two factor authentication.
- `security_manage_active_directory` - (Optional) Whether the team can manage global active directory. Only relevant for the DDI product.
- `dhcp_manage_dhcp` - (Optional) Whether the team can manage DHCP. Only relevant for the DDI product.
- `dhcp_view_dhcp` - (Optional) Whether the team can view DHCP. Only relevant for the DDI product.
- `ipam_manage_ipam` - (Optional) Whether the team can manage IPAM. Only relevant for the DDI product.
- `ipam_view_ipam` - (Optional) Whether the team can view IPAM. Only relevant for the DDI product.

» Attributes Reference

All of the arguments listed above are exported as attributes, with no additions.

» ns1_user

Provides a NS1 User resource. Creating a user sends an invitation email to the user's email address. This can be used to create, modify, and delete users. The credentials used must have the `manage_users` permission set.

» Example Usage

```
resource "ns1_team" "example" {
  name = "Example team"

  # Optional IP whitelist
  ip_whitelist = ["1.1.1.1","2.2.2.2"]

  dns_view_zones       = false
  account_manage_users = false
}

resource "ns1_user" "example" {
  name     = "Example User"
  username = "example_user"
  email    = "user@example.com"
  teams    = ["${ns1_team.example.id}"]
}
```

» Permissions

A user will inherit permissions from the teams they are assigned to. If a user is assigned to a team and also has individual permissions set on the user, the individual permissions will be overridden by the inherited team permissions. In a future release, setting permissions on a user that is part of a team will be explicitly disabled.

When a user is removed from all teams completely, they will inherit whatever permissions they had previously. If a user is removed from all their teams, it will probably be necessary to run `terraform apply` a second time to update the users permissions from their old team permissions to new user-specific permissions.

See the NS1 API docs for an overview of permission semantics.

» Argument Reference

The following arguments are supported:

- `name` - (Required) The free form name of the user.
- `username` - (Required) The users login name.
- `email` - (Required) The email address of the user.
- `notify` - (Required) Whether or not to notify the user of specified events. Only `billing` is available currently.
- `teams` - (Required) The teams that the user belongs to.
- `ip_whitelist` - (Optional) The IP addresses to whitelist for this key.
- `ip_whitelist_strict` - (Optional) Sets exclusivity on this IP whitelist.
- `dns_view_zones` - (Optional) Whether the user can view the accounts zones.
- `dns_manage_zones` - (Optional) Whether the user can modify the accounts zones.
- `dns_zones_allow_by_default` - (Optional) If true, enable the `dns_zones_allow` list, otherwise enable the `dns_zones_deny` list.
- `dns_zones_allow` - (Optional) List of zones that the user may access.
- `dns_zones_deny` - (Optional) List of zones that the user may not access.
- `data_push_to_datafeeds` - (Optional) Whether the user can publish to data feeds.
- `data_manage_datasources` - (Optional) Whether the user can modify data sources.
- `data_manage_datafeeds` - (Optional) Whether the user can modify data feeds.
- `account_manage_users` - (Optional) Whether the user can modify account users.
- `account_manage_payment_methods` - (Optional) Whether the user can modify account payment methods.
- `account_manage_plan` - (Optional) Whether the user can modify the account plan.
- `account_manage_teams` - (Optional) Whether the user can modify other teams in the account.
- `account_manage_apikeys` - (Optional) Whether the user can modify account apikeys.
- `account_manage_account_settings` - (Optional) Whether the user can modify account settings.
- `account_view_activity_log` - (Optional) Whether the user can view activity logs.
- `account_view_invoices` - (Optional) Whether the user can view invoices.
- `monitoring_manage_lists` - (Optional) Whether the user can modify notification lists.

- `monitoring_manage_jobs` - (Optional) Whether the user can modify monitoring jobs.
- `monitoring_view_jobs` - (Optional) Whether the user can view monitoring jobs.
- `security_manage_global_2fa` - (Optional) Whether the user can manage global two factor authentication.
- `security_manage_active_directory` - (Optional) Whether the user can manage global active directory. Only relevant for the DDI product.
- `dhcp_manage_dhcp` - (Optional) Whether the user can manage DHCP. Only relevant for the DDI product.
- `dhcp_view_dhcp` - (Optional) Whether the user can view DHCP. Only relevant for the DDI product.
- `ipam_manage_ipam` - (Optional) Whether the user can manage IPAM. Only relevant for the DDI product.

» Attributes Reference

All of the arguments listed above are exported as attributes, with no additions.