

» **scaleway__bootscript**

Use this data source to get the ID of a registered Bootscript for use with the `scaleway_server` resource.

» **Example Usage**

```
data "scaleway_bootscript" "debug" {
  architecture = "arm"
  name_filter  = "Rescue"
}
```

» **Argument Reference**

- `architecture` - (Optional) any supported Scaleway architecture, e.g. `x86_64`, `arm`
- `name_filter` - (Optional) Regexp to match Bootscript name by
- `name` - (Optional) Exact name of desired Bootscript

» **Attributes Reference**

`id` is set to the ID of the found Bootscript. In addition, the following attributes are exported:

- `architecture` - architecture of the Bootscript, e.g. `arm` or `x86_64`
- `organization` - uuid of the organization owning this Bootscript
- `public` - is this a public bootscript
- `boot_cmd_args` - command line arguments used for booting
- `dtb` - path to Device Tree Blob detailing hardware information
- `initrd` - URL to initial ramdisk content
- `kernel` - URL to used kernel

» **scaleway__image**

Use this data source to get the ID of a registered Image for use with the `scaleway_server` resource.

» Example Usage

```
data "scaleway_image" "ubuntu" {
  architecture = "arm"
  name         = "Ubuntu Precise"
}

resource "scaleway_server" "base" {
  name  = "test"
  image = "${data.scaleway_image.ubuntu.id}"
  type  = "C1"
}
```

» Argument Reference

- `architecture` - (Required) any supported Scaleway architecture, e.g. `x86_64`, `arm`
- `name_filter` - (Optional) Regexp to match Image name by
- `name` - (Optional) Exact name of desired Image

» Attributes Reference

`id` is set to the ID of the found Image. In addition, the following attributes are exported:

- `architecture` - architecture of the Image, e.g. `arm` or `x86_64`
- `organization` - uuid of the organization owning this Image
- `public` - is this a public bootscript
- `creation_date` - date when image was created

» `scaleway__ip`

Provides IPs for servers. This allows IPs to be created, updated and deleted. For additional details please refer to API documentation.

» Example Usage

```
resource "scaleway_ip" "test_ip" {}
```

» Argument Reference

The following arguments are supported:

- **server** - (Optional) ID of server to associate IP with
- **reverse** - (Optional) Reverse DNS of the IP

» Attributes Reference

The following attributes are exported:

- **id** - ID of the new resource
- **ip** - IP of the new resource
- **server** - ID of the associated server resource
- **reverse** - reverse DNS setting of the IP resource

» Import

Instances can be imported using the `id`, e.g.

```
$ terraform import scaleway_ip.jump_host 5faef9cd-ea9b-4a63-9171-9e26bec03dbc
```

» `scaleway__server`

Provides servers. This allows servers to be created, updated and deleted. For additional details please refer to API documentation.

» Example Usage

```
resource "scaleway_server" "test" {
  name   = "test"
  image  = "5faef9cd-ea9b-4a63-9171-9e26bec03dbc"
  type   = "VC1M"

  volume {
    size_in_gb = 20
    type       = "l_ssd"
  }
}
```

» Argument Reference

The following arguments are supported:

- **name** - (Required) name of server
- **image** - (Required) base image of server
- **type** - (Required) type of server
- **bootscript** - (Optional) server bootscript
- **tags** - (Optional) list of tags for server
- **enable_ipv6** - (Optional) enable ipv6
- **dynamic_ip_required** - (Optional) make server publicly available
- **security_group** - (Optional) assign security group to server
- **volume** - (Optional) attach additional volumes to your instance (see below)
- **public_ipv6** - (Read Only) if **enable_ipv6** is set this contains the ipv6 address of your instance
- **state** - (Optional) allows you to define the desired state of your server. Valid values include (**stopped**, **running**)
- **state_detail** - (Read Only) contains details from the scaleway API the state of your instance

Field **name**, **type**, **tags**, **dynamic_ip_required**, **security_group** are editable.

» Volume

You can attach additional volumes to your instance, which will share the lifetime of your **scaleway_server** resource.

Warning: Using the **volume** attribute does not modify the System Volume provided default with every **scaleway_server** instance. Instead it adds additional volumes to the server instance.

The **volume** mapping supports the following:

- **type** - (Required) The type of volume. Can be "l_ssd"
- **size_in_gb** - (Required) The size of the volume in gigabytes.

» Attributes Reference

The following attributes are exported:

- **id** - id of the new resource
- **private_ip** - private ip of the new resource
- **public_ip** - public ip of the new resource

» Import

Instances can be imported using the id, e.g.

```
$ terraform import scaleway_server.web 5faef9cd-ea9b-4a63-9171-9e26bec03dbc
```

» scaleway__security__group

Provides security groups. This allows security groups to be created, updated and deleted. For additional details please refer to API documentation.

» Example Usage

```
resource "scaleway_security_group" "test" {  
  name          = "test"  
  description    = "test"  
  enable_default_security = true  
}
```

» Argument Reference

The following arguments are supported:

- **name** - (Required) name of security group
- **description** - (Required) description of security group
- **enable_default_security** - (Optional) default: true. Add default security group rules

Field **name**, **description** are editable.

» Attributes Reference

The following attributes are exported:

- **id** - id of the new resource

» Import

Instances can be imported using the id, e.g.

```
$ terraform import scaleway_security_group.test 5faef9cd-ea9b-4a63-9171-9e26bec03dbc
```

» `scaleway_security_group_rule`

Provides security group rules. This allows security group rules to be created, updated and deleted. For additional details please refer to API documentation.

» Example Usage

```
resource "scaleway_security_group" "test" {
  name          = "test"
  description    = "test"
}

resource "scaleway_security_group_rule" "smtp_drop_1" {
  security_group = "${scaleway_security_group.test.id}"

  action      = "accept"
  direction   = "inbound"
  ip_range    = "0.0.0.0/0"
  protocol     = "TCP"
  port        = 25
}
```

» Argument Reference

The following arguments are supported:

- `security_group` - (Required) the security group which should be associated with this rule
- `action` - (Required) action of rule (`accept`, `drop`)
- `direction` - (Required) direction of rule (`inbound`, `outbound`)
- `ip_range` - (Required) ip_range of rule
- `protocol` - (Required) protocol of rule (`ICMP`, `TCP`, `UDP`)
- `port` - (Optional) port of the rule

Fields `action`, `direction`, `ip_range`, `protocol`, `port` are editable.

» Attributes Reference

The following attributes are exported:

- `id` - id of the new resource

» **scaleway__volume**

Provides volumes. This allows volumes to be created, updated and deleted. For additional details please refer to API documentation.

» **Example Usage**

```
resource "scaleway_server" "test" {
  name      = "test"
  image     = "aecaed73-51a5-4439-a127-6d8229847145"
  type      = "C2S"
  volumes = ["${scaleway_volume.test.id}"]
}

resource "scaleway_volume" "test" {
  name      = "test"
  size_in_gb = 20
  type      = "l_ssd"
}
```

» **Argument Reference**

The following arguments are supported:

- **name** - (Required) name of volume
- **size_in_gb** - (Required) size of the volume in GB
- **type** - (Required) type of volume
- **server** - (Read Only) the **scaleway_server** instance which has this volume mounted right now

» **Attributes Reference**

The following attributes are exported:

- **id** - id of the new resource

» **Import**

Instances can be imported using the id, e.g.

```
$ terraform import scaleway_volume.test 5faef9cd-ea9b-4a63-9171-9e26bec03dbc
```

» `scaleway__volume__attachment`

This allows volumes to be attached to servers.

Warning: Attaching volumes requires the servers to be powered off. This will lead to downtime if the server is already in use.

» Example Usage

```
resource "scaleway_server" "test" {
  name   = "test"
  image  = "aecaed73-51a5-4439-a127-6d8229847145"
  type   = "C2S"
}

resource "scaleway_volume" "test" {
  name      = "test"
  size_in_gb = 20
  type      = "l_ssd"
}

resource "scaleway_volume_attachment" "test" {
  server = "${scaleway_server.test.id}"
  volume = "${scaleway_volume.test.id}"
}
```

» Argument Reference

The following arguments are supported:

- `server` - (Required) id of the server
- `volume` - (Required) id of the volume to be attached

» Attributes Reference

The following attributes are exported:

- `id` - id of the new resource